

9/4.8.2

Courbes polaires d'équation $r = f(\theta)$

COMMENT UTILISER LE PROGRAMME

Voici le listing du programme :

```
1000 ' -----
1010 ' Trace d'une courbe polaire
1020 ' d'equation r=f(theta)
1030 ' avec recherche de l'echelle optimale
1040 ' -----
1050 '
1060 MODE 2
1070 PRINT"Etude d'une fonction polaire"
1080 PRINT"-----"
1090 PRINT
1100 PRINT"Tapez "
1110 PRINT" 2050 r="
1120 PRINT"suivi de la fonction a etudier"
1130 PRINT
1140 PRINT"Par exemple : "
1150 PRINT" 2050 r=sin(t)"
1160 PRINT
1170 PRINT"Tapez ensuite"
1180 PRINT" run 1230"
1190 PRINT"pour effectuer le trace"
1200 PRINT
1210 STOP
1220 '
1230 DEG
1240 '
1250 ' - - - - -
1260 ' Saisie de l'intervalle d'etude
1270 ' - - - - -
1280 '
1290 PRINT"Entrez l'intervalle d'etude en t : "
1300 INPUT" Minimum : ";mi
1310 INPUT" Maximum : ";ma
1320 '
1330 ' - - - - -
```

```

1340 ' Calcul du min et du max en X et en Y
1350 ' -----
1360 '
1370 ux=-1E+15 'Maximum en X
1380 uy=-1E+15 'Maximum en Y
1390 lx=1E+15 'Minimum en X
1400 ly=1E+15 'Minimum en Y
1410 FOR t=mi TO ma STEP (ma-mi)/100
1420   GOSUB 2000
1430   IF x<lx THEN lx=x
1440   IF y<ly THEN ly=y
1450   IF x>ux THEN ux=x
1460   IF y>uy THEN uy=y
1470 NEXT t
1480 '
1490 ' -----
1500 ' Trace de la courbe
1510 ' -----
1520 '
1530 CLS
1540 LOCATE 16,1
1550 PRINT"Trace de courbes polaires d'equation r=f(teta)"
1560 t=mi : GOSUB 2000
1570 cty=345-(291*ly/(ly-uy)):ctx=54-(531*lx/(ux-lx))
1580 MOVE 50,50 : DRAW 589,50 : DRAW 589,349 : DRAW 50,349 :
DRAW 50,50
1590 MOVE 54,54 : DRAW 585,54 : DRAW 585,345 : DRAW 54,345 :
DRAW 54,54
1600 FOR t=mi TO ma STEP (ma-mi)/100
1610   GOSUB 2000
1620   IF t=mi THEN MOVE ctx+(531/(ux-lx))*x,399-(cty+(Y*291
)/(ly-uy))
1630   IF x<>mi THEN DRAW ctx+(531/(ux-lx))*x,399-(cty+(Y*29
1)/(ly-uy))
1640 NEXT t
1650 a$=INKEY$:IF a$="" THEN 1650
1660 CLS
1670 END
2000 '
2010 ' -----
2020 ' Equation a etudier
2030 ' -----
2040 '
2050 r=SIN(t)*SIN(3*t)
2060 x=r*COS(t)
2070 y=r*SIN(t)
2080 RETURN

```

Le fonctionnement du programme est très similaire à celui du précédent. L'équation doit être entrée ligne 2050 sous la forme :

$$R = f(t)$$

La saisie des bornes du domaine d'étude et le tracé sont réalisés par une instruction

RUN 1230

Remarque :


Si vous étudiez des fonctions sinusoidales, le domaine d'étude doit être exprimé en degrés.

Pour quitter le programme, appuyez sur une touche quelconque du clavier.

Exemple :

Voici le résultat obtenu pour l'équation

$$R = \text{SIN}(t) * \text{SIN}(3t)$$



LE PROGRAMME EN DÉTAIL

Le programme est très similaire au précédent. Nous n'y reviendrons que partiellement.

Dans ce programme, il faut calculer deux facteurs d'échelle : un en X et un en Y. En effet, la variable est ici t, alors que la représentation se fait dans un repère XY. Le procédé employé est le même si ce n'est qu'il concerne maintenant l'ordonnée et l'abscisse (lignes 1370 à 1470).

- Le tracé de la courbe est également semblable au tracé précédent.
- Remarquez cependant lignes 2060 et 2070 que les coordonnées X et Y sont calculées à partir du cosinus et du sinus de l'équation.